

Инструменты и методы отладки формул

В ЭТОЙ ГЛАВЕ...

- Что такое отладка формул
- Обнаружение и исправление наиболее распространенных ошибок в формулах
- Средства аудита Excel

К сожалению, ошибки в программах не редкость. Довольно часто они встречаются и при создании формул Excel. В данной главе описаны наиболее распространенные ошибки в формулах, а также представлены средства и методы, которые призваны помочь в создании формул, корректно выполняющих свою работу.

Что такое отладка формул

Термин *отладка* означает процесс идентификации и исправления ошибок в компьютерных программах. Конечно, формулы Excel не являются компьютерными программами. Вместе с тем им присущи проблемы того же рода, что и компьютерным программам. Если созданная формула не работает так, как предполагалось, необходимо обнаружить источник проблемы и устранить его.

Конечной целью разработки электронной таблицы является получение правильного результата. Добиться этого в случае простых рабочих листов, как правило, нетрудно. Но по мере увеличения размера и сложности рабочих листов обеспечить достоверность получаемых результатов становится все сложнее.

Внесение изменений в рабочий лист, даже весьма незначительных, может вызвать эффект домино, который приведет к возникновению ошибок в других ячейках. Скажем, можно случайно ввести значение в ячейку, в которой до этого уже содержалась формула. Зачастую такая простая ошибка существенно влияет на другие формулы рабочего листа, а сама ошибка может быть обнаружена спустя большой промежуток времени после ее возникновения (или вообще может быть никогда не обнаружена).



ИССЛЕДОВАНИЕ ОШИБОК НА РАБОЧЕМ ЛИСТЕ

Использование электронных таблиц может представлять для компании определенные риски. Весьма соблазнительно просто полагать, что электронная таблица всегда дает точные результаты. Если эти результаты используются для принятия важных решений, возвращение формулами точных и поддающихся интерпретации результатов приобретает особое значение.

Было проведено немало исследований, объектом которых являлись ошибки в электронных таблицах. В целом в ходе этих исследований было обнаружено, что от 20 до 40 процентов всех электронных таблиц содержат ошибки того или иного рода. Если вам интересны подобные исследования, посетите сайт организации European Spreadsheet Risk Interest Group по следующему адресу:

<http://eusprig.org>

Проблемы, возникающие в формулах, и способы их устранения

Ошибки в формулах можно разбить на шесть общих категорий.

- ▶ **Синтаксические ошибки.** Не соблюдены синтаксические правила написания формул. Например, формула содержит непарные скобки, или функция имеет неверное количество аргументов.
- ▶ **Логические ошибки.** Формула не возвращает ошибку, однако она логически неверна и возвращает неправильный результат.
- ▶ **Неверные ссылки.** С логикой в формуле все в порядке, однако в ней используется неверная ссылка на ячейку (например, формула СУММ может охватывать не все данные, которые нужно просуммировать).
- ▶ **Семантические ошибки.** Примером семантической ошибки может служить неверно написанное имя функции. Такую функцию Excel пытается интерпретировать как имя и возвращает ошибку #ИМЯ?.
- ▶ **Циклические ссылки.** Циклическая ссылка встречается, когда формула прямо или косвенно ссылается на ячейку, в которую она введена. Иногда циклические ссылки создаются намеренно, но в большинстве случаев они приводят к возникновению ошибки.
- ▶ **Ошибка при вводе формулы массива.** При вводе (или редактировании) формулы массива необходимо использовать комбинацию клавиш <Ctrl+Shift+Enter>. Если этого не сделать, Excel не распознает такую формулу как формулу массива. Если вам повезет, Excel сообщит об ошибке; в худшем случае вы получите неправильный результат, даже не подозревая об этом.
- ▶ **Ошибка незавершенных вычислений.** Формулы могут не вычисляться в полном объеме из-за встроенных в Excel процедур оптимизации производительности.

Для гарантии полного вычисления формул используйте комбинацию клавиш <Ctrl+Alt+Shift+F9>.

Обычно проще всего обнаружить и устранить синтаксические ошибки. В большинстве случаев хорошо видно, в каком месте была допущена ошибка. Например, Excel не позволит ввести формулу с недостающими скобками. Другие синтаксические ошибки также обычно приводят к появлению в ячейке сообщения об ошибке.

В данном разделе описаны некоторые общие проблемы, возникающие в формулах, а также даны советы по их обнаружению и устранению.

Непарные скобки

В формуле каждой открывающей (левой) скобке должна соответствовать закрывающая (правая) скобка. Обычно, если в формуле отсутствует необходимая скобка, Excel не позволит ввести такую формулу. Исключением из данного правила являются простые формулы, использующие функцию. Например, если ввести следующую формулу (с недостающей закрывающей скобкой), Excel воспримет ее и самостоятельно добавит недостающую скобку:

```
=СУММ(А1:500
```

Формула может содержать равное количество открывающих и закрывающих скобок, но при этом они могут быть расставлены неправильно. Для примера рассмотрим следующую формулу, преобразующую текстовую строку таким образом, чтобы ее первый символ стал прописным, а все остальные — строчными. Эта формула содержит пять пар скобок, расставленных должным образом:

```
=ПРОПИСН(ЛЕВСИМВ(А1)) & ПРАВСИМВ(СТРОЧН(А1); ДЛСТР(А1) - 1)
```

Следующая формула также имеет пять пар скобок, однако расставлены они неверно. В результате получается синтаксически правильная формула, которая возвращает неверный результат:

```
=ПРОПИСН(ЛЕВСИМВ(А1) & ПРАВСИМВ(СТРОЧН(А1); ДЛСТР(А1) - 1))
```

Часто неправильная расстановка скобок приводит к возникновению синтаксической ошибки, которая сопровождается сообщением о том, что введено слишком много или слишком мало аргументов функции.



Совет


Excel может оказать вам помощь при работе с недостающими скобками. Если при редактировании формулы подвести указатель к скобке и немного выждать, Excel примерно на одну секунду выделит данную скобку и соответствующую ей парную полужирным начертанием. К тому же соответствующие друг другу вложенные скобки отображаются одним цветом, при этом каждый уровень вложенности имеет свой цвет.



АВТОКОРРЕКЦИЯ ФОРМУЛ

При вводе формулы, содержащей синтаксическую ошибку, Excel пытается определить проблему и предлагает ее решение в окне, приведенном ниже (чтобы оно появилось,

нужно набрать формулу вручную; если вставить формулу из буфера обмена, окно автокоррекции не появится).



Будьте осторожны, принимая предлагаемые исправления, так как они не всегда оказываются верными. Для примера введем следующую формулу (с недостающими скобками):

```
=СРЗНАЧ (СУММ (A1 : A12; СУММ (B1 : B12) )
```

Excel предложит представленное ниже исправление для формулы.

```
=СРЗНАЧ (СУММ (A1 : A12; СУММ (B1 : B12) ) )
```

Если вы без раздумий примете такие изменения, то получите синтаксически корректную формулу, но не ту, которая на самом деле предполагалась и имеет следующий вид:

```
=СРЗНАЧ (СУММ (A1 : A12) ; СУММ (B1 : B12) )
```

Ячейки заполнены знаками

Ячейка может отображать серию знаков решетки (#) по одной из следующих причин.

- Ширины столбца недостаточно для размещения отформатированного числового значения. Чтобы исправить такую ситуацию, можно либо расширить ячейку, либо использовать другой числовой формат.
- Ячейка содержит формулу, возвращающую некорректную дату или время. Например, Excel не поддерживает даты до 1900 года и отрицательные значения времени. Попытка отображения таких значений приводит к заполнению ячейки символами решетки. Увеличение ширины ячейки не исправляет данную ошибку.

Пустые ячейки на самом деле не пусты

Возможно, вы думаете, что при нажатии клавиши <пробел> содержимое ячейки уничтожается. На самом деле при этом вставляется невидимый символ пробела, что не эквивалентно уничтожению содержимого ячейки.

Например, следующая формула возвращает количество заполненных ячеек в диапазоне A1:A10. Если содержимое некоторых ячеек данного диапазона удалено с помощью клавиши пробела, эти ячейки будут засчитаны и формула вернет неверный результат:

```
=СЧЁТЗ (A1 : A10)
```

Если формула не игнорирует пустые ячейки, проверьте, действительно ли они пусты. Чтобы найти ячейки, содержащие только символы пробела, выполните следующие действия.

1. Нажмите клавиши <Ctrl+F> для открытия диалогового окна **Найти и заменить**.
2. В поле **Найти** введите символ пробела.
3. Установите флажок **Ячейка целиком**.
4. Щелкните на кнопке **Найти все**.

Если хотя бы одна ячейка содержит только символы пробела, в нижней части окна **Найти и заменить** отобразится их список.

Лишние символы пробела

Если существуют формулы, сравнивающие текст, будьте особенно внимательны: некоторые строки могут содержать лишние пробелы. Проблема наличия лишних пробелов особенно часто возникает при импорте данных из внешних источников.

Excel автоматически удаляет замыкающие пробелы при вводе данных вручную, однако это не относится к текстовым строкам. Глядя на ячейку, невозможно уверенно сказать, содержатся ли в ней замыкающие пробелы.

Для удаления ведущих и замыкающих пробелов, а также последовательностей пробелов в середине текстовой строки предназначена функция СЖПРОБЕЛЫ.

На рис. 22.1 показан некоторый текст в столбце А. В ячейке В1 содержится приведенная ниже формула, скопированная в другие ячейки столбца.

=СЖПРОБЕЛЫ(А1)=А1

Эта формула возвращает значение ЛОЖЬ, если ячейка содержит ведущий, замыкающий пробел или несколько пробелов, следующих один за другим. В данном примере после слова Собака есть замыкающий пробел, хотя это и не видно.

	А	В	С
1	Морж	ИСТИНА	
2	Кот	ИСТИНА	
3	Собака	ЛОЖЬ	
4	Корова	ИСТИНА	
5	Цыпленок	ИСТИНА	
6			
7			
8			
9			

Рис. 22.1. Использование формулы для обнаружения ячеек, содержащих лишние пробелы

Формула возвращает ошибку

Формула может вернуть один из следующих кодов ошибки:

- #ДЕЛ/0;
- #Н/Д;
- #ИМЯ?;
- #ПУСТО!;
- #ЧИСЛО!;
- #ССЫЛКА!;
- #ЗНАЧ!.

В последующих разделах рассмотрены возможные проблемы, вызывающие данные ошибки.



Совет

Excel позволяет выбрать способ отображения ошибок при выводе листа на печать. Для доступа к данному средству откройте диалоговое окно Параметры страницы, щелкнув на небольшой кнопке в правом нижнем углу области параметров Разметка страницы⇒Параметры страницы. На вкладке Лист открывшегося диалогового окна можно выбрать один из способов вывода ошибок на печать — как на экране, в виде пустых ячеек, в виде прочерков или в виде кода ошибки #Н/Д.



ТРАССИРОВКА ОШИБОК

Зачастую ошибка в одной ячейке обусловлена ошибкой в другой ячейке (которая используется в формуле). Для трассировки формулы и поиска источника ошибки можно воспользоваться командой Формулы⇒Зависимости формул⇒Проверка наличия ошибки⇒Источник ошибки. Excel отобразит на экране стрелками путь от кода ошибки к ее источнику.

	A	B	C	D	E	F	G	H
1		Кол-во изделий	Цена	Всего				
2	Север	82	1,2	98,4				
3	Юг	396	1,8	712,8				
4	Восток	473	1,9	#ЗНАЧ!				
5	Запад	437	1,1	480,7				
6	Итого	1388		#ЗНАЧ!				
7								
8						Выручка	#ЗНАЧ!	
9						Затраты	1450	
10						Прибыль	#ЗНАЧ!	
11								

После идентификации ошибки с помощью команды Формулы⇒Зависимости формул⇒Убрать стрелки⇒Убрать стрелки можно очистить рабочий лист от этих стрелок.

Ошибка #ДЕЛ/0!

Деление на ноль является недопустимой операцией. Если попытаться ее выполнить, Excel отобразит хорошо знакомое значение ошибки #ДЕЛ/0!.

Так как Excel считает значение пустой ячейки равным нулю, данная ошибка будет получена и при делении на отсутствующее значение. Это стандартная ситуация при создании формул для еще не введенных значений (рис. 22.2). В ячейке D2 находится следующая формула, которая скопирована вниз:

= (C2-B2) / C2

	А	В	С	Д	Е
1	Изделие	Прошлый год	Текущий год	Изменение	
2	Виджет 4	175	162	-7,43%	
3	Виджет 4s	156	147	-5,77%	
4	Виджет 5	198	202	2,02%	
5	Виджет 5с	144	166	15,28%	
6	Виджет 6		149	#ДЕЛ/0!	
7					
8					
9					
10					
11					

Рис. 22.2. Ошибка #ДЕЛ/0! возникает в случае отсутствия значений в столбце С



На сайте

Рабочая книга с приведенными в этой главе примерами содержится в файле `commission calc.xlsx`.

Эта формула находит прирост производства изделий в процентах по отношению к предыдущему году. Поскольку новое изделие в прошлом году не выпускалось, данные для него отсутствуют, и поэтому формула возвращает код ошибки #ДЕЛ/0!.

Чтобы избежать отображения ошибки, воспользуйтесь функцией ЕСЛИ, которая проверяет наличие пустых ячеек в столбце С.

`=ЕСЛИ(В2=0; ""; (С2-В2)/С2)`

Данная формула выводит пустую строку, если ячейка В2 пустая или содержит нуль; в противном случае будет выведено вычисленное значение.

Еще одно решение заключается в том, чтобы использовать функцию ЕСЛИОШИБКА для проверки возникновения *любой* ситуации ошибки. Следующая формула отобразит пустую строку в случае получения ошибки любого типа:

`=ЕСЛИОШИБКА((С2-В2)/С2); ""`

Функция ЕСЛИОШИБКА впервые появилась в версии Excel 2007. Чтобы обеспечить совместимость с предыдущими версиями, необходимо использовать следующую формулу:

`=ЕСЛИ(ЕОШИБКА((С2-В2)/С2); ""; (С2-В2)/С2)`

Ошибка #Н/Д

Ошибка #Н/Д (нет данных) возникает, если в одной из ячеек, на которую ссылается формула, отображается значение #Н/Д.



Совет

Некоторые пользователи предпочитают явно указывать на отсутствие данных в ячейке, вводя в ней формулу `=ЗНАЧ()` или значение #Н/Д (что означает "недоступно"). При этом становится совершенно ясно, что данные недоступны, а не удалены случайно.

Ошибка #Н/Д возникает также тогда, когда какой-либо функция поиска (ГПР, ПРОСМОТР, ВПР или ПОИСКПОЗ) не удается найти соответствие.

Если вы захотите отображать вместо #Н/Д пустую строку, используйте в формуле функцию ЕСЛИНД:

```
=ЕСЛИНД(ВПР(A1;C1:F50;4;ЛОЖЬ);"")
```



Новинка

Функция ЕСЛИНД введена в Excel 2013. Для обеспечения совместимости с предыдущими версиями необходимо изменить формулу:

```
=ЕСЛИ(ЕСНД(ВПР(A1;C1:F50;4;ЛОЖЬ));";";ВПР(A1;C1:F50;4;ЛОЖЬ))
```

Третьей распространенной причиной возникновения ошибки #Н/Д являются различия в количестве элементов, содержащихся в различных компонентах формулы массива. Приведенная ниже в качестве примера формула массива используется для поиска наибольшего значения в столбцах А и В, где столбец А1 — Север, а столбец В — Июнь.

```
{=МАКС((A1:A100="Север")*(B1:B99="Июнь")*(C1:C100))}
```

В первом и третьем массивах поиск осуществляется в 100 строках, тогда как второй массив включает всего лишь 99 строк. В силу этого различия данная формула возвратит значение #Н/Д.



Ссылка

Формулы массива рассматриваются в главе 14.

Ошибка #ИМЯ?

Ошибка #ИМЯ? возникает в следующих случаях.

- Формула содержит отсутствующее имя ячейки или диапазона.
- Формула содержит текст, который Excel интерпретирует как неопределенное имя. Например, неправильное написание имени функции приводит к отображению ошибки #ИМЯ?.
- Формула содержит текст, не заключенный в кавычки.
- Формула содержит ссылку на диапазон, в которой опущено двоеточие между адресами ячеек.
- Формула использует функцию рабочего листа, определенную в надстройке, а данная надстройка не установлена.



Примечание

В Excel часто возникают проблемы с именованными диапазонами. Если удалить имя ячейки или диапазона, используемое в формуле, то формула и в дальнейшем будет применять это имя, хоть оно больше не определено. Как результат, формула отображает ошибку #ИМЯ?. Если вы считаете, что Excel автоматически преобразует имена в соответствующие ссылки на ячейки, то глубоко заблуждаетесь. На самом деле в Excel даже не реализован механизм преобразования имен, используемых в формуле, в эквивалентные ссылки на ячейки.

Ошибка #ПУСТО!

Данная ошибка возникает, когда формула пытается использовать пересечение двух диапазонов, которые в действительности не имеют общих ячеек. Оператором пересече-

ния областей является пробел между ссылками. Например, следующая формула возвращает значение #ПУСТО!, так как два указанных диапазона не имеют общих ячеек:

```
=СУММ(B5:B14 A16:F20)
```

Представленная ниже формула не возвратит значение #ПУСТО!. Вместо этого она отобразит содержимое ячейки B9, являющейся пересечением двух указанных диапазонов.

```
=СУММ(B5:B14 A9:F16)
```

Ошибка #ПУСТО! возникает также, если случайно опустить оператор в формуле. Например, в следующей формуле пропущен второй оператор:

```
= A1+A2 A3
```

Ошибка #ЧИСЛО!

Ошибка #ЧИСЛО! возникает в одном из следующих случаев.

- Для функции, принимающей числовой аргумент, задан аргумент другого типа. Например, передан аргумент \$1000, когда ожидается 1000.
- В функцию передан недопустимый аргумент. Например, ошибку #ЧИСЛО! вернет следующая формула:

```
=КОРЕНЬ(-1)
```
- Используемая итерации функция не может вычислить результат. К функциям, использующим итерации, относятся, например, функции ВСД и СТАВКА.
- Формула возвращает слишком большое или слишком малое значение. Excel поддерживает числовые величины в пределах от $-1E-307$ до $1E+307$.

Ошибка #ССЫЛКА!

Ошибка #ССЫЛКА! возникает в случае использования формулой ошибочной ссылки на ячейку. Данная ошибка может возникнуть в следующих ситуациях.

- Ячейка, на которую ссылается формула, была удалена. Например, представленная далее формула отобразит ошибку #ССЫЛКА!, если удалена строка 1, столбец А или столбец В.

```
=A1/B1
```
- Удален рабочий лист с ячейкой, на которую ссылается формула. Например, следующая формула возвращает ошибку #ССЫЛКА!, если удален Лист2.

```
=Лист2!A1
```
- Формула скопирована в новое место, где относительные ссылки на ячейки становятся недействительными. Например, если скопировать следующую формулу из ячейки А2 в ячейку А1, формула вернет ошибку #ССЫЛКА!, так как в ней будет присутствовать ссылка на несуществующую ячейку.

```
=A1-1
```
- Ячейка была вырезана (с помощью команды Главная⇒Буфер обмена⇒Вырезать или комбинации клавиш <Ctrl+X>), а затем вставлена в ячейку, на которую ссылается формула. Формула отобразит ошибку #ССЫЛКА!.

Ошибка #ЗНАЧ!

Ошибка #ЗНАЧ! весьма распространена и встречается в следующих ситуациях.

- В качестве аргумента функции использованы данные несоответствующего типа, или же формула пытается выполнить операцию, используя некорректные данные. Например, формула, складывающая значение с текстовой строкой, вернет ошибку #ЗНАЧ!.
- В качестве аргумента функции применяется диапазон, тогда как аргументом должно быть одиночное значение.
- Пользовательская функция на языке VBA не вычислена. В некоторых версиях Excel вставка или перемещение листов может приводить к возникновению данной ошибки. Для принудительного пересчета воспользуйтесь комбинацией клавиш <Ctrl+Alt+F9>.
- Пользовательская функция пытается выполнить недопустимую операцию. Например, она не может изменить среду Excel или внести изменения в другие ячейки.
- При вводе формулы массива вы забыли нажать комбинацию клавиш <Ctrl+Shift+Enter>.



ОБРАЩАЙТЕ ВНИМАНИЕ НА ЦВЕТ ФОНА ЯЧЕЙКИ

При редактировании содержимого ячейки, включающей формулу, Excel использует цветовое выделение для обозначения ссылок на ячейки и диапазоны. При этом на рабочем листе контурами такого же цвета выделяются соответствующие ячейки и диапазоны. Таким образом, вы имеете возможность визуально просмотреть данные, используемые формулой.

Цветные контуры можно изменять, модифицируя ссылки на ячейку или диапазон. Для изменения используемой ссылки перетащите границу контура или маркер выделения (который находится в правом нижнем углу контура). Использовать данный прием, как правило, легче, чем изменять формулу вручную.

Проблемы с абсолютными и относительными ссылками

Как уже отмечалось в главе 2, ссылки на ячейки могут быть относительными (например, A1), абсолютными (например, \$A\$1) и смешанными (например, A\$1 или \$A1). Тип ссылки становится существенным только при копировании формулы в другие ячейки.

Довольно часто ошибки возникают вследствие использования относительных ссылок вместо абсолютных. На рис. 22.3 ячейка C1 содержит ставку налога, используемую в формулах столбца C. Формула в ячейке C4 имеет следующий вид:

=B4*(1+\$C\$1)

Обратите внимание на то, что ссылка на ячейку C1 является абсолютной. При копировании формулы в другие ячейки столбца C она продолжает ссылаться на ячейку C1. Если бы ссылка на ячейку C1 была относительной, формула возвращала бы неверный результат.

C4		=B4*(1+\$C\$1)	
	A	B	C
1		Ставка налога:	7,25%
2			
3	Изделие	Цена	Цена+налог
4	A-544	149,95	160,82
5	B-102	79,95	85,75
6	R-099	32,00	34,32
7	R-123	32,00	34,32
8			
9			
10			

Рис. 22.3. Формулы в ячейках C4 : C7 используют абсолютную ссылку на ячейку C1

Проблемы приоритета операторов

В Excel установлены строгие правила относительно порядка выполнения математических операций. В табл. 22.1 приведены операторы с указанием их приоритета (чем меньше число, тем раньше выполняется оператор). Например, из данной таблицы видно, что умножение имеет более высокий приоритет, чем сложение, поэтому умножение будет выполнено раньше.

Таблица 22.1. Приоритет операторов в формулах Excel

Символ	Оператор	Приоритет
-	Смена знака (обращение)	1
%	Процент	2
^	Степень	3
* и /	Умножение и деление	4
+ и -	Сложение и вычитание	5
&	Объединение текста	6
=, <, > и <>	Сравнение	7

Если у вас возникают сомнения в вопросах подобного рода (или вы просто хотите сделать выражение более понятным), используйте круглые скобки. Это будет гарантировать выполнение операций в том порядке, который вам нужен. Например, следующая формула умножает ячейку A1 на ячейку A2, а затем добавляет к результату 1. Умножение выполняется раньше, так как данная операция имеет более высокий приоритет:

=1+A1*A2

Ниже представлена более наглядная запись данной формулы. Использование скобок не обязательно, но с ними порядок выполнения действий становится очевидным.

=1+(A1*A2)

Обратите внимание на то, что операторы отрицания и вычитания выглядят одинаково, что может приводить к путанице. Рассмотрим две формулы:

```
=-3^2
=0-3^2
```

Первая формула, как и ожидается, возвращает значение 9, вторая — значение -9. Возведение в квадрат всегда дает положительный результат, поэтому возникает вопрос: как же программа получила отрицательное значение?

Дело в том, что в первой формуле знак “минус” является оператором отрицания и имеет более высокий приоритет. Во втором случае знак “минус” — это оператор вычитания, имеющий приоритет ниже, чем оператор возведения в степень. Таким образом, значение 3 возводится в квадрат, а затем результат вычитается из нуля, что и дает отрицательное значение.



Примечание

Интерпретация оператора отрицания в Excel немного необычна. Другие электронные таблицы (например, Lotus 1-2-3 или Quattro Pro) возвращают -9 в обеих формулах. Более того, при использовании языка VBA в Excel также возвращается результат -9 для этих выражений.

Формулы не вычислены

Если на рабочем листе применяются пользовательские функции, написанные на VBA, может случиться так, что у вызывающей их формулы возникают проблемы с пересчетом и формула возвращает неверный результат. Предположим, например, что функция VBA возвращает числовой формат указанной ячейки. Если изменить текущий формат, функция по-прежнему будет отображать предыдущий формат. Это объясняется тем, что операция изменения формата не запускает процедуру вычисления.

Чтобы вынудить пересчет одной формулы, выделите ячейку и нажмите клавиши <F2> и <Enter>. Для принудительного пересчета всех формул используйте комбинацию клавиш <Ctrl+Alt+F9>. Создание функций VBA рассматривается в части IV.

Реальные и отображаемые значения

Вы можете столкнуться с ситуацией, когда кажется, будто суммирование выполняется с ошибкой. На рис. 22.4 показан рабочий лист, содержащий в каждой из ячеек диапазона B2 : B4 следующую формулу:

```
=1/3
```

Ячейка B5 содержит следующую формулу:

```
=СУММ(B2:B4)
```

Все эти ячейки отформатированы для отображения чисел с тремя знаками после запятой. Можно предположить, что формула в ячейке B5 дает неправильный результат (казалось бы, результат должен быть равен 0,999). Однако данная формула возвращает правильный результат, так как она использует реальные значения из диапазона B2 : B4, а не отображаемые на экране.

Можно дать указание Excel использовать в вычислениях отображаемые значения, установив флажок **Задать точность как на экране** в разделе **Дополнительно** ⇒ При

пересчете этой книги диалогового окна параметров программы. Этот флажок применяется к активной рабочей книге.

	A	B	C	D
1				
2				
3		0,333		
4		0,333		
5		0,333		
6		1,000		
7				
8				
9				
10				

Рис. 22.4. Простая демонстрация “неправильного” сложения чисел



Внимание

Используйте флажок **Задать точность**, как на экране с осторожностью, четко понимая механизм его работы. Его действие распространяется не только на формулы, но и непосредственно на значения, вводимые в ячейки. Например, если ячейка содержит значение 4,68 и отображается без десятичных знаков (т.е. как 5), установка данного флажка приведет к преобразованию числа 4,68 в 5,00. Данное изменение необратимо, поэтому восстановить прежнее значение при снятии флажка **Задать точность**, как на экране невозможно. Гораздо лучшим решением является использование функции **ОКРУГЛ**, которая округляет значения до заданного количества знаков после запятой (см. главу 10). За все долгие годы работы с программой Excel у авторов книги никогда не возникала потребность в установке флажка **Задать точность**, как на экране.

Ошибки при работе с числами с плавающей запятой

Компьютеры по самой своей природе не могут обеспечить бесконечную точность. Excel хранит числа в двоичном представлении (используя для них 8 байт), которое обеспечивает точность до 15 десятичных знаков. Некоторые числа нельзя точно выразить с использованием восьми байт, поэтому они аппроксимируются.

Чтобы продемонстрировать, как данное положение может привести к возникновению проблем, введем в ячейке A1 следующую формулу:

$= (5,1 - 5,2) + 1$

Результатом данного выражения должно быть 0,9. Однако, если отформатировать данную ячейку для отображения 15 десятичных знаков, отобразится значение 0,899999999999999. Причиной этого является то, что сначала выполняется действие в скобках, а промежуточный результат сохраняется в двоичном коде после аппроксимации. Затем формула добавляет к данному значению 1, и ошибка аппроксимации распространяется на конечный результат.

В большинстве случаев данный тип ошибки не представляет особой проблемы. Но, если результат такой формулы проверяется логическим оператором, возможны проблемы. Например, следующая формула (если в ячейке A1 находится предыдущая формула) возвращает значение **ЛОЖЬ**, хотя должна возвращать значение **ИСТИНА**:

$=A1 = ,9$

Одним из возможных способов устранения данной ошибки является использование функции ОКРУГЛ. Следующая формула, благодаря сравнению значения в ячейке A1, округленного до одного десятичного знака, возвращает значение ИСТИНА:

=ОКРУГЛ(A1;1)=0,9

Ниже приведен еще один пример так называемой “проблемы точности”. Попробуйте ввести следующую формулу:

=(1,333-1,233)-(1,334-1,234)

Эта формула должна вернуть нуль, однако на самом деле вернет 2,220446E-16 (число, очень близкое к нулю).

Если ввести эту формулу в ячейку A1, то следующая формула вернет текст Не нуль:

=ЕСЛИ(A1=0;"Нуль";"Не нуль")

Одним из способов избежать проблемы “округления” до нуля является использование формулы, подобной приведенной ниже.

=ЕСЛИ(ABS(A1)<1E-6;"Нуль";"Не нуль")

В этой формуле используется оператор “меньше” для сравнения модуля числа с очень малым числом. В результате возвращается значение Нуль.

“Висячие” ссылки

После открытия рабочей книги вы можете увидеть сообщение, извещающее о существовании ссылок на другие источники данных (рис. 22.5). Это сообщение иногда отображается даже в случае, когда рабочая книга вообще не содержит связанных формул.




Рис. 22.5. Excel спрашивает, не хотите ли вы обновить ссылки в рабочей книге

Прежде всего попробуйте использовать команду **Файл**⇒**Сведения**⇒**Изменить связи с файлами**. В открывшемся диалоговом окне **Изменение связей** выделяйте каждую из ссылок и щелкайте на кнопке **Разорвать связь**. Если этот прием не решает проблему, “висячая” ссылка могла быть вызвана в результате использования ошибочного имени. Выберите команду **Формулы**⇒**Определенные имена**⇒**Диспетчер имен**. В открывшемся диалоговом окне просмотрите список имен. Если встретите имя, ссылающееся на значение ошибки #ССЫЛКА!, удалите его. Окно диспетчера имен содержит кнопку **Фильтр**, с помощью которой можно отфильтровать имена с ошибками.



Ссылка

“Висячие” ссылки могут образовываться при копировании рабочего листа, содержащего имена. Более детальную информацию об именах см. в главе 3.

Ошибки логических значений

В ячейку можно ввести слова ЛОЖЬ и ИСТИНА, тем самым создав соответствующие логические значения. Несмотря на всю простоту этих имен, Excel не всегда адекватно трактует их.

На рис. 22.6 показан рабочий лист с тремя логическими значениями, введенными в ячейки A1:A3. Одновременно в ячейки A5:A7 введены формулы, суммирующие эти логические значения. Как видите, все эти формулы возвращают разные результаты.

	A	B	C
1	ИСТИНА		
2	ИСТИНА		
3	ЛОЖЬ		
4			
5	2	=A1+A2+A3	
6	0	=СУММ(A1:A3)	
7	2	{=СУММ(--A1:A3)}	
8	-2	=VBASUMM(A1:A3)	
9			
10			
11			

Рис. 22.6. На этом рабочем листе продемонстрировано отсутствие единообразия при суммировании логических значений в Excel

Формула в ячейке A5 использует операторы сложения. Суммой всех трех ячеек является число 2. Вывод следующий: Excel трактует значение ИСТИНА как единицу, а значение ЛОЖЬ — как нуль.

В ячейке A6 для суммирования используется функция рабочего листа СУММ. Как видите, возвращаемое ею значение — нуль. Другими словами, функция СУММ игнорирует логические значения. В то же время можно заставить ее трактовать логические значения как числа, используя формулу массива. Введите следующую формулу и нажмите <Ctrl+Shift+Enter>:

```
=СУММ(A1:A3*1)
```

Еще более усложняет ситуацию то, что если логические значения передаются в функцию СУММ как константы, то она возвращает неверное значение 2:

```
=СУММ(ИСТИНА;ИСТИНА;ЛОЖЬ)
```

И наконец, третий случай (ячейка A8). Несмотря на то что язык VBA тесно интегрирован с Excel, иногда создается впечатление, будто два приложения совершенно “не понимают” друг друга. Рассмотрим простую функцию VBA, которая суммирует значения в диапазоне (ее текст приведен ниже). Эта функция возвращает результат -2!

```
Function VBASUM(rng)
    Dim cell As Range
    VBASUM = 0
    For Each cell In rng
```

```

VBASUM = VBASUM + cell.Value
Next cell
End Function

```

Все дело в том, что язык VBA трактует значение ИСТИНА как -1 , а значение ЛОЖЬ — как ноль.

Мораль такова: принимайте в расчет непоследовательность работы Excel с логическими значениями и будьте предельно внимательны при суммировании диапазонов, содержащих их.

Ошибки циклических ссылок

Циклической ссылкой является формула с прямой или косвенной ссылкой на ячейку, содержащую эту формулу. Вопросы отслеживания циклических ссылок подробно рассмотрены ниже.

Средства аудита Excel

В Excel предусмотрены инструменты, призванные оказывать помощь в нахождении ошибок формул. В данном разделе описаны встроенные в Excel средства аудита.

Нахождение ячеек определенного типа

Диалоговое окно **Выделение группы ячеек** (рис. 22.7) позволяет задать тип ячеек, которые программа Excel должна выделить на экране. Чтобы открыть это диалоговое окно, выберите команду Главная⇒Редактирование⇒Найти и выделить⇒Выделить группу ячеек.




Рис. 22.7. Диалоговое окно **Выделить группу ячеек**



Примечание

Если при открытии окна **Выделение группы ячеек** выделен диапазон из нескольких ячеек, команда будет применена только к ним. Если же выбрана всего одна ячейка, команда применяется ко всему рабочему листу.

Диалоговое окно **Выделить группу ячеек** используется для выделения групп ячеек определенного типа, что часто облегчает поиск ошибок. Например, если установить переключатель **формулы**, Excel выделит все ячейки, которые содержат формулы. Если вы обнаружите, что одна из таких ячеек осталась невыделенной, то, скорее всего, это означает, что ранее содержащаяся в ней формула впоследствии была заменена значением.

Просмотр формул

Во многих случаях понять назначение незнакомой рабочей книге легче в режиме просмотра формул, а не результатов их вычисления. Чтобы перейти в этот режим, выберите команду **Формулы** ⇒ **Зависимости формул** ⇒ **Показать формулы**. Возможно, прежде чем выполнять эту команду, вы захотите создать для текущей рабочей книги второе окно. Использование двух окон обеспечивает возможность одновременного просмотра как значений, так и формул. Для этого выберите команду **Вид** ⇒ **Окно** ⇒ **Новое окно**.



Совет

Для переключения между режимами просмотра значений и формул также можно использовать комбинацию клавиш <Ctrl+'> (клавиша '<', соответствующая символу обратного удара, находится над клавишей <Tab>).

На рис. 22.8 приведен пример одновременного отображения рабочей книги в двух окнах. В верхнем окне используется режим обычного просмотра (результаты вычисления формул), а в нижнем отображаются сами формулы. Для одновременного просмотра двух окон вам пригодится команда **Вид** ⇒ **Рядом**, которая позволяет синхронизировать прокрутку их содержимого.

Верхнее окно (commission calc.xlsx - Excel) показывает таблицу с результатами вычисления:

Торговый представитель	Прошлый месяц	Текущий месяц	Изменение	Процент изменения	Цель достигнута?	Сумма комиссионных
Муррей	101 233	108 444	7 211	7,1%	ИСТИНА	7 049
Наис	120 933	108 434	(12 499)	-10,3%	ЛОЖЬ	5 964
Лефи	139 832	165 901	26 069	18,6%	ИСТИНА	10 794
Лаки	98 323	100 083	1 760	1,8%	ЛОЖЬ	5 505
Скарфейс	78 322	79 923	1 601	2,0%	ЛОЖЬ	4 396
Итого	538 643	562 785	24 142	4,5%		33 697
Средний размер комиссионных						6,0%

Нижнее окно (commission calc.xlsx2 - Excel) показывает ту же таблицу, но с формулами:

Торговый представитель	Прошлый месяц	Текущий месяц	Изменение	Процент изменения	Цель достигнута?	Сумма комиссионных
Муррей	101233	108444	=C6-B6	=D6/B6	=E6>=B\$2	=С6*ЕСЛИ(F6;S\$3;S\$5)
Наис	120933	108434	=C7-B7	=D7/B7	=E7>=B\$2	=С7*ЕСЛИ(F7;S\$3;S\$5)
Лефи	139832	165901	=C8-B8	=D8/B8	=E8>=B\$2	=С8*ЕСЛИ(F8;S\$3;S\$5)
Лаки	98323	100083	=C9-B9	=D9/B9	=E9>=B\$2	=С9*ЕСЛИ(F9;S\$3;S\$5)
Скарфейс	78322	79923	=C10-B10	=D10/B10	=E10>=B\$2	=С10*ЕСЛИ(F10;S\$3;S\$5)
Итого	=СУММ(B6:B10)	=СУММ(C6:C10)	=СУММ(D6:D10)	=D11/B11		=СУММ(G6:G10)
Средний размер комиссионных			=G11/C11			

Рис. 22.8. Отображение формул (нижнее окно) и результатов их вычисления (верхнее окно)

При включенном режиме просмотра формул Excel выделяет цветом ячейки, используемые формулой в активной ячейке. В примере на рис. 22.9 активна ячейка G10. Четыре ячейки, используемые этой формулой, подсвечены в обоих окнах.



НАДСТРОЙКА INQUIRE

Некоторые дистрибутивы Excel 2016 предоставляют удобную надстройку Inquire, предназначенную для аудита рабочих листов. Для установки надстройки Inquire выберите команду Файл⇒Параметры. В диалоговом окне Параметры Excel откройте категорию Надстройки и выберите в верхнем списке пункт Inquire. Выберите в раскрывающемся списке Надстройки COM и щелкните на кнопке Перейти.

В диалоговом окне Надстройки для модели компонентных объектов (COM) установите флажок Inquire и щелкните на кнопке ОК. Теперь надстройка будет автоматически загружаться при запуске Excel. Если надстройки Inquire в списке нет, значит, установленная версия Excel не поддерживает ее.

Инструменты надстройки Inquire доступны на отдельной одноименной вкладке ленты. Надстройку можно использовать для решения следующих задач.

- Сравнение версий рабочих книг.
- Анализ рабочей книги на предмет потенциальных проблем и несогласованностей.
- Отображение интерактивного окна диагностики.
- Визуализация связей между рабочей книгой и рабочими листами.
- Очистка избыточного форматирования ячеек.
- Управление паролями.

Отслеживание связей ячеек

Для того чтобы понять, как можно отследить взаимосвязи ячеек, необходимо ознакомиться с двумя концепциями.

- **Влияющие ячейки.** Ячейки, от которых зависит результат формулы, находящейся в другой ячейке. Термин *прямое влияние* относится к ячейкам, используемым непосредственно в формуле. Если в формуле используются не сами влияющие ячейки, а их значения, то говорят о *непрямом (косвенном) влиянии*.
- **Зависимые ячейки.** Это ячейки с формулами, зависящие от текущей ячейки. Множество зависимых ячеек состоит из всех ячеек, содержащих формулы, прямо или косвенно зависящие от текущей ячейки

Рассмотрим, например, формулу, введенную в ячейку A4:

=СУММ(A1:A3)

Ячейка A4 имеет три влияющих ячейки (A1, A2 и A3), каждая из которых имеет одну зависимую ячейку — A4. В данном случае все зависимости прямые.

Определение ячеек, влияющих на формулу, помогает разобраться в том, почему данная формула работает неправильно. Знание того, какие ячейки с формулами зависят от определенной ячейки, также полезно. Например, если вы собираетесь удалить формулу, информация о наличии зависимых ячеек позволит избежать потенциальных ошибок.

Идентификация влияющих ячеек

Определить ячейки, используемые формулой в активной ячейке, можно несколькими способами.

- ▶ **Нажмите клавишу <F2>**. Ячейки, используемые непосредственно в формуле, будут обведены цветными контурами. Цвет контура соответствует цвету ссылки на ячейку в формуле. Этот прием ограничен только ячейками, содержащимися на одном листе с формулой.
- ▶ **Откройте диалоговое окно Выделить группу ячеек** (с помощью команды Главная⇒Редактирование⇒Найти и выделить⇒Выделить группу ячеек). Установите переключатель Влияющие ячейки, а затем установите, в зависимости от своих намерений, подчиненный переключатель Только непосредственно или На всех уровнях. Щелкните на кнопке ОК, и Excel выделит все ячейки, влияющие на данную формулу. Этот прием также ограничен только ячейками, содержащимися на одном листе с формулой.
- ▶ **Нажмите комбинацию клавиш <Ctrl+[>** для выделения на активном рабочем листе всех ячеек, прямо влияющих на данную ячейку.
- ▶ **Нажмите клавиши <Ctrl+Shift+[>**, чтобы выделить на активном рабочем листе все ячейки, прямо или косвенно влияющие на данную ячейку.

	A	B	C	D	E	F	G
1	Комиссионные	5,5%	Обычный размер комиссионных				
2	Целевой показатель	5%	Рост объема продаж по сравнению с прошлым месяцем				
3	Повышенные комиссионные	6,5%	Размер комиссионных при достижении целевого показателя				
4							
5	Торговый представитель	Прошлый месяц	Текущий месяц	Изменение	Процент изменения	Цель достигнута?	Сумма комиссионных
6	Мюррей	101 233	108 444	7 211	7,1%	ИСТИНА	7 049
7	Наклс	120 933	108 434	(12 499)	-10,3%	ЛОЖЬ	5 964
8	Лефти	139 832	165 901	26 069	18,6%	ИСТИНА	10 784
9	Лакки	98 323	100 083	1 760	1,8%	ЛОЖЬ	5 505
10	Скарфейс	78 322	79 923	1 601	2,0%	ЛОЖЬ	4 396
11	Итого	538 643	562 785	24 142	4,5%		33 697
12							
13	Средний размер комиссионных		6,0%				
14							
15							
16							

Рис. 22.9. На рабочем листе стрелками обозначены ячейки, влияющие на формулу

- Выберите команду **Формулы**⇒**Зависимости формул**⇒**Влияющие ячейки**. Excel отобразит на рабочем листе стрелки, указывающие на зависимые ячейки. Чтобы скрыть стрелки, выберите команду **Формулы**⇒**Зависимости формул**⇒**Убрать стрелки**. На рис. 22.9 показан рабочий лист с трассировкой ячеек, влияющих на текущую ячейку (в данном случае ею является ячейка C13).

Нахождение зависимых ячеек

Найти формулы, зависящие от отдельной ячейки, можно несколькими способами.

- Откройте диалоговое окно **Выделить группу ячеек** (с помощью команды **Главная**⇒**Редактирование**⇒**Найти и выделить**⇒**Выделить группу ячеек**). Установите переключатель **Зависимые ячейки**, а затем установите, в зависимости от своих намерений, подчиненный переключатель **Только непосредственно** или **На всех уровнях**. Щелкните на кнопке **ОК**, и Excel выделит все ячейки, зависящие от текущей. Этот прием ограничен только ячейками, находящимися на одном листе с формулой.
- Нажмите комбинацию клавиш **<Ctrl+>** для выделения на активном рабочем листе всех напрямую зависимых ячеек.
- Нажмите комбинацию клавиш **<Ctrl+Shift+>**, чтобы выделить на активном рабочем листе все зависимые (напрямую или косвенно) ячейки.
- Выберите команду **Формулы**⇒**Зависимости формул**⇒**Зависимые ячейки**. Excel отобразит на рабочем листе стрелки, указывающие на зависимые от текущей ячейки. Чтобы скрыть стрелки, выберите команду **Формулы**⇒**Зависимости формул**⇒**Убрать стрелки**.

Отслеживание ошибочных значений

Если формула отображает код ошибки, Excel поможет идентифицировать ячейку, вызвавшую появление этой проблемы. Зачастую ошибка в формуле вызвана ошибкой во влияющей на нее ячейке. Сделайте активной ячейку, содержащую ошибку, после чего выберите команду **Формулы**⇒**Зависимости формул**⇒**Проверка наличия ошибок**⇒**Источник ошибки**. Excel отобразит на рабочем листе стрелки, указывающие на источник ошибки.

Исправление ошибок циклической ссылки

При случайном вводе формулы, содержащей циклическую ссылку, Excel открывает окно предупреждения, отображает в строке состояния сообщение **Циклическая ссылка** (с адресом ячейки), а на рабочем листе рисует стрелки, позволяющие идентифицировать источник проблемы. Если сложно найти источник проблем, воспользуйтесь командой **Формулы**⇒**Зависимости формул**⇒**Проверка наличия ошибок**⇒**Циклические ссылки**. Меню этой команды содержит список всех ячеек, вовлеченных в циклическую ссылку. Начните с выделения первой из ячеек списка, а затем переходите к следующим до тех пор, пока источник проблемы не будет обнаружен.

Фоновая проверка ошибок

Возможно, вас порадует тот факт, что в Excel предусмотрено средство автоматической проверки ошибок. Фоновая проверка включается и отключается с помощью флажка **Включить фоновый поиск ошибок** вкладки **Формулы** диалогового окна параметров Excel (рис. 22.11). В разделе **Правила контроля ошибок** можно определить, какие типы ошибок следует выявлять. Для этого достаточно установить соответствующие флажки.




Рис. 22.10. Excel может проверить формулы, чтобы найти потенциальные ошибки

Когда включена фоновая проверка ошибок, Excel постоянно пересчитывает рабочий лист, в том числе и его формулы. При обнаружении потенциальной ошибки Excel помещает в левом верхнем углу «подозрительной» ячейки значок в виде небольшого треугольника. Если ячейка активизирована, отображается смарт-тег. После щелчка на нем вам предлагается несколько вариантов дальнейших действий. На рис. 22.12 показаны варианты действий, которые доступны после щелчка на смарт-теге ячейки, содержащей ошибку #ДЕЛ/0!. Доступные действия изменяются в зависимости от типа ошибки.

Во многих случаях можно игнорировать ошибку, выбрав команду **Пропустить ошибку**. Выбор данной команды исключает ячейку из списка ошибок. Можно снова отобразить все пропущенные ошибки и для этого необходимо щелкнуть на кнопке **Сброс пропущенных ошибок**, находящейся на вкладке **Формулы** диалогового окна параметров Excel (см. рис. 22.11).

	A	B	C	D	E	F
1	Изделие	Прошлый год	Текущий год	Изменение		
2	Виджет 4	175	162	-7,43%		
3	Виджет 4s	156	147	-5,77%		
4	Виджет 5	198	202	2,02%		
5	Виджет 5с	144	166	15,28%		
6	Виджет 6			#ДЕЛ/0!		
7						
8						
9						
10						
11						
12						
13						
14						
15						

Рис. 22.11. После щелчка на смарт-теге открывается список возможных действий

Воспользуйтесь командой **Формулы**⇒**Зависимости формул**⇒**Проверка наличия ошибок** для последовательного отображения ячеек с потенциально возможными ошибками (подобно тому, как это делает программа проверки орфографии). Диалоговое окно Поиск ошибок показано на рис. 22.12. Обратите внимание, что это окно *немодальное* (когда оно открыто, вы по-прежнему имеете доступ к содержимому рабочего листа).




Рис. 22.12. Использование диалогового окна поиска ошибок для последовательного просмотра найденных программой потенциальных ошибок



Внимание

Важно понимать, что средство проверки ошибок несовершенно. Нельзя считать, что рабочий лист полностью лишен ошибок только на основании того, что программа Excel не выявила источников потенциальной опасности! Кроме того, имейте в виду, что данное средство не сможет отследить такую часто встречающуюся ошибку, как замена формулы значением.

Проверка формул

В Excel предусмотрено еще одно специальное средство, которое позволяет просмотреть различные части вложенной формулы, представленные в порядке ее вычисления.

Чтобы использовать эту возможность, выделите содержащую формулу ячейку, а затем выберите команду **Формулы**⇒**Зависимости формул**⇒**Вычислить формулу**. Откроется диалоговое окно, показанное на рис. 22.13.




Рис. 22.13. Средство вычисления формул Excel позволяет просматривать результаты промежуточных этапов вычислений

Щелкните на кнопке **ВЫЧИСЛИТЬ**, чтобы отобразить результат вычисления выражения формулы. После каждого щелчка на кнопке вычисляется подчеркнутая часть формулы. На первый взгляд этот инструмент может показаться несколько сложным, однако если вы поработаете с ним некоторое время, то поймете принцип его действия и по достоинству оцените его возможности.

Excel предлагает еще один способ вычисления части формулы.

1. Выделите ячейку, содержащую формулу.
2. Нажмите клавишу <F2>, чтобы перейти в режим редактирования содержимого ячейки.
3. С помощью мыши выделите часть формулы, которую хотите вычислить. Для этого также можно воспользоваться клавишами со стрелками, удерживая нажатой клавишу <Shift>.
4. Нажмите клавишу <F9>.

Подсвеченная часть формулы отображает полученный результат. Далее можно таким же образом вычислить остальные части этой же формулы или нажать клавишу <Esc> и выйти из режима редактирования, вернув формулу в ее первоначальное состояние.



Внимание

Будьте осторожны при использовании этого приема, так как случайное нажатие клавиши <Enter> (вместо <Esc>) приведет к нежелательному изменению формулы, в результате чего в ней будут использоваться вычисленные значения.